
  Description


Library for building distributed systems that are scalable. It handles the distribution of
processes within a cluster of nodes while providing a globally synchronized process registry.
ProcessHub takes care of starting, stopping and monitoring processes in the cluster.
It scales automatically when cluster is updated and handles network partitions.
ProcessHub provides different configuration options to define whether it operates in a
decentralized or leader-based architecture. The default distribution strategy is
decentralized and based on consistent hashing, where each node in the cluster is considered
equal. Alternatively, you can configure ProcessHub to use a centralized load balancer
strategy that relies on a leader node to make distribution decisions based on real-time
cluster metrics.
ProcessHub is eventually consistent
ProcessHub is built with scalability and availability in mind. 
Most of the operations are asynchronous and non-blocking. It can guarantee eventual consistency.
The system may not be in a consistent state at all times, 
but it will eventually converge to a consistent state.


  
    
  
  Features


Main features include:
	Automatically or manually distribute processes within a cluster of nodes.
	Distributed and synchronized process registry for fast process lookups.
	Process monitoring and automatic restart on failures (Supervisor).
	Process state handover on process migration.
	Provides different strategies out of the box to handle: 	Process migrations.
	Process distribution.
	Process synchronization.
	Process replication.
	Cluster partitioning.


	Hooks for triggering events on specific actions and extend the functionality.
	Automatic hub cluster forming and healing when nodes join or leave the cluster.
	Customizable and extendable to alter the default behavior of the system by implementing custom hook handlers and strategies.


  
    
  
  Installation


	Add process_hub to your list of dependencies in mix.exs:
 def deps do
   [
     {:process_hub, "~> 0.4.0-beta"}
   ]
 end

	Run mix deps.get to fetch the dependencies.    

	Add ProcessHub to your application's supervision tree:
 defmodule MyApp.Application do
   use Application

   def start(_type, _args) do
     children = [
       ProcessHub.child_spec(%ProcessHub{hub_id: :my_hub})
     ]

     opts = [strategy: :one_for_one, name: MyApp.Supervisor]
     Supervisor.start_link(children, opts)
   end
 end


  It is possible to start multiple hubs under the same supervision tree, each
  with a unique :hub_id.
  By doing so, each hub will have its own cluster of processes. 
  All hubs will be independent of each other.
  For example we can start two separate hubs with different configurations.

  
    
  
  Dynamic process creation


Dynamically create 2 distributed processes under the hub :my_hub. These processes are
started asynchronously by default and are monitored by the hub.
iex> ProcessHub.start_children(:my_hub, [
  %{id: "process1", start: {MyProcess, :start_link, [nil]}},
  %{id: "process2", start: {MyProcess, :start_link, [nil]}}
])
{:ok, :start_initiated}

  
    
  
  Static process creation


Start the hub with 2 child specs. The hub will start the processes when it boots up.
child_specs = [
  %{
    id: "my_process_1",
    start: {MyProcess, :start_link, [nil]}
  },
  %{
    id: "my_process_2",
    start: {MyProcess, :start_link, [nil]}
  }
]

# Start under the supervision tree.
ProcessHub.child_spec(%ProcessHub{
  hub_id: :my_hub,
  child_specs: child_specs
})

  
    
  
  Process lookup


Query the whole registry for all processes under the hub :my_hub:
iex> ProcessHub.process_list(:my_hub, :global)
[
  {"my_process_1", [node_two@host: #PID<23772.233.0>]},
  {"my_process_2", [node_one@user: #PID<0.250.0>]}
]
Query processes by child_id:
iex> ProcessHub.child_lookup(:my_hub, "my_process_1")
{
  %{id: "my_process_1", start: {MyProcess, :start_link, [nil]}},
  [node_two@host: #PID<0.228.0>]
}
Find pid of a process by child_id:
iex> ProcessHub.get_pid(:my_hub, :my_process_1)
#PID<0.228.0>
See the documentation for more guides.

  
    
  
  Contributing


Contributions are welcome and appreciated. If you have any ideas, suggestions, or bugs to report,
please open an issue or a pull request on GitHub.


  

    
Getting started
    

In this guide we will go through the basic usage of ProcessHub and how to integrate 
it in your application.
To get more familiar with the public API of ProcessHub, please refer to the ProcessHub module.

  
    
  
  Installation


In order to use ProcessHub in your application, you need to add it as a dependency.
Once you have added it as a dependency, you need to start the ProcessHub master supervisor.
It is highly recommended to start the ProcessHub supervisor under your application 
supervision tree.
	Add process_hub to your list of dependencies in mix.exs:
 def deps do
     [
         {:process_hub, "~> 0.4.0-beta"}
     ]
 end

	Pull the dependency:
 mix deps.get


	Start the ProcessHub supervisor under your application supervision tree:
 defmodule MyApp.Application do
     use Application

     def start(_type, _args) do
         children = [
             ProcessHub.child_spec(%ProcessHub{hub_id: :my_hub})
         ]

         opts = [strategy: :one_for_one, name: MyApp.Supervisor]
         Supervisor.start_link(children, opts)
     end
 end


Hub ID
Each node in the cluster that starts a ProcessHub instance and shares the same :hub_id will 
form a hub cluster.
It is possible to start multiple hubs under the same supervision tree with
different configuration options, in this case, each hub must have a unique :hub_id.


  
    
  
  Example usage


The following example shows how to start 2 elixir nodes, connect them and start processes
under the ProcessHub cluster. This demonstrates how the processes are distributed in the cluster.
Note: The examples below assume that the ProcessHub is already started under the
supervision tree. If not please refer to the Installation section.
Note: Make sure you have a GenServer module called MyProcess defined in your project.
defmodule MyProcess do
    use GenServer

    def start_link(_) do
        GenServer.start_link(__MODULE__, nil)
    end

    def init(_) do
        {:ok, nil}
    end
end

  
    
  
  Node 1


Start the first node with the following command:
iex --name node1@127.0.0.1 --cookie mycookie -S mix

# Run the following in the iex console to start 5 processes under the hub.
iex> ProcessHub.start_children(:my_hub, [
...>    %{id: :process1, start: {MyProcess, :start_link, [nil]}},
...>    %{id: :process2, start: {MyProcess, :start_link, [nil]}},
...>    %{id: :process3, start: {MyProcess, :start_link, [nil]}},
...>    %{id: :process4, start: {MyProcess, :start_link, [nil]}},
...>    %{id: :process5, start: {MyProcess, :start_link, [nil]}}
...> ])
{:ok, :start_initiated}

# Check the started processes by running the command below.
iex> ProcessHub.process_list(:my_hub, :global)
[
  process1: ["node1@127.0.0.1": #PID<0.256.0>],
  process2: ["node1@127.0.0.1": #PID<0.257.0>],
  process3: ["node1@127.0.0.1": #PID<0.258.0>],
  process4: ["node1@127.0.0.1": #PID<0.259.0>],
  process5: ["node1@127.0.0.1": #PID<0.260.0>]
]

  
    
  
  Node 2


We will use this node to connect to the first node and see how the processes are
automatically distributed.
Start the second node.
iex --name node2@127.0.0.1 --cookie mycookie -S mix

# Connect the second node to the first node.
iex> Node.connect(:"node1@127.0.0.1")
true

# Check the started procsses by running the command below and
# see how some of the processes are distributed to the second node.
iex> ProcessHub.process_list(:my_hub, :global)
[
  process4: ["node2@127.0.0.1": #PID<0.285.0>],
  process3: ["node2@127.0.0.1": #PID<0.284.0>],
  process1: ["node2@127.0.0.1": #PID<0.283.0>],
  process2: ["node1@127.0.0.1": #PID<23045.257.0>],
  process5: ["node1@127.0.0.1": #PID<23045.260.0>]
]


  

    
ProcessRegistry
    

The ProcessRegistry is a global registry that keeps track of all the processes in the system. 
It is used to look up a process by its name, and to register a process with a name.
It's recommended to interact with ProcessRegistry through the main ProcessHub API module.
By default ProcessRegistry is using pubs/subs mechanism to notify subscribers about changes in the registry. This implementation can be replaced with ProcessHub.Strategy.Synchronization.Gossip or
custom implementation.
ProcessRegistry can also be used to store metadata about the processes, such as tags or other information. 
This can be useful to group processes, query them by tags or distribute data among them and trigger any custom logic.
The metadata is synchronized across the cluster, so all nodes will have the same information about the processes.
We will skip the boiler plate code for starting the child processes and focus on the registry itself.

  
    
  
  Querying the whole registry


We will start of by querying the whole registry with ProcessHub.process_list/2 function.
This function allows us to query the whole registry or a specific hub and specify if we want to include remote processes or not.
If we choose to exclude remote processes, the node names wont be included in the returned list.
We can switch between :global and :local mode by passing the second argument to the function.
The ProcessHub.process_list/2function is the most efficient way to query the whole registry because
the data is coming from the local registry and no network calls are made.
iex> ProcessHub.process_list(:my_hub, :global)
[
  my_process_1: [two@anuar: #PID<23772.233.0>],
  my_process_2: [one@anuar: #PID<0.250.0>],
  my_process_3: [one@anuar: #PID<0.253.0>],
  my_process_4: [one@anuar: #PID<0.256.0>],
  my_process_5: [one@anuar: #PID<0.259.0>],
  my_process_6: [one@anuar: #PID<0.262.0>],
  my_process_7: [one@anuar: #PID<0.265.0>],
  my_process_8: [two@anuar: #PID<23772.254.0>],
  my_process_9: [one@anuar: #PID<0.271.0>],
  my_process_10: [one@anuar: #PID<0.274.0>]
]

iex> ProcessHub.process_list(:my_hub, :local)
[
  my_process_2: #PID<0.250.0>,
  my_process_3: #PID<0.253.0>,
  my_process_4: #PID<0.256.0>,
  my_process_5: #PID<0.259.0>,
  my_process_6: #PID<0.262.0>,
  my_process_7: #PID<0.265.0>,
  my_process_9: #PID<0.271.0>,
  my_process_10: #PID<0.274.0>
]
Another similar option is ProcessHub.registry_dump/1 which dumps the whole registry into a map
including metadata stored in the registry.
iex> ProcessHub.registry_dump(:my_hub)
%{
  my_process_1: {%{id: :my_process_1, start: {MyProcess, :start_link, [nil]}},
   [two@anuar: #PID<23772.233.0>], %{tag: "Some tag"}},
  my_process_2: {%{id: :my_process_2, start: {MyProcess, :start_link, [nil]}},
   [one@anuar: #PID<0.250.0>], %{custom_data: "Some data"}},
  my_process_3: {%{id: :my_process_3, start: {MyProcess, :start_link, [nil]}},
   [one@anuar: #PID<0.253.0>], %{}},
  my_process_4: {%{id: :my_process_4, start: {MyProcess, :start_link, [nil]}},
   [one@anuar: #PID<0.256.0>], %{}},
  my_process_5: {%{id: :my_process_5, start: {MyProcess, :start_link, [nil]}},
   [one@anuar: #PID<0.259.0>], %{}},
  my_process_6: {%{id: :my_process_6, start: {MyProcess, :start_link, [nil]}},
   [one@anuar: #PID<0.262.0>], %{}},
  my_process_7: {%{id: :my_process_7, start: {MyProcess, :start_link, [nil]}},
   [one@anuar: #PID<0.265.0>], %{}},
  my_process_8: {%{id: :my_process_8, start: {MyProcess, :start_link, [nil]}},
   [two@anuar: #PID<23772.254.0>], %{}},
  my_process_9: {%{id: :my_process_9, start: {MyProcess, :start_link, [nil]}},
   [one@anuar: #PID<0.271.0>], %{}},
  my_process_10: {%{id: :my_process_10, start: {MyProcess, :start_link, [nil]}},
   [one@anuar: #PID<0.274.0>], %{}}
}
There's also a third option ProcessHub.which_children/2 but is highly discouraged to use and may 
be deprecated in the future.
This function does network calls to all nodes in the cluster to get the list of processes and
also may lead to memory and performance issues when used with large nuber of processes.
It's using the Supervisor.which_children/1 function under the hood which has stated in the 
documentation: Note that calling this function when supervising a large number of children
  under low memory conditions can cause an out of memory exception.
Here is an example of how to use ProcessHub.which_children/2 function:
iex> ProcessHub.which_children(:my_hub, [:global])
[
  one@anuar: [
    {:my_process_10, #PID<0.274.0>, :worker, [MyProcess]},
    {:my_process_9, #PID<0.271.0>, :worker, [MyProcess]},
    {:my_process_7, #PID<0.265.0>, :worker, [MyProcess]},
    {:my_process_6, #PID<0.262.0>, :worker, [MyProcess]},
    {:my_process_5, #PID<0.259.0>, :worker, [MyProcess]},
    {:my_process_4, #PID<0.256.0>, :worker, [MyProcess]},
    {:my_process_3, #PID<0.253.0>, :worker, [MyProcess]},
    {:my_process_2, #PID<0.250.0>, :worker, [MyProcess]}
  ],
  two@anuar: [
    {:my_process_8, #PID<23772.254.0>, :worker, [MyProcess]},
    {:my_process_1, #PID<23772.233.0>, :worker, [MyProcess]}
  ]
]

iex> ProcessHub.which_children(:my_hub, [:local])
{:one@anuar,
 [
   {:my_process_10, #PID<0.274.0>, :worker, [MyProcess]},
   {:my_process_9, #PID<0.271.0>, :worker, [MyProcess]},
   {:my_process_7, #PID<0.265.0>, :worker, [MyProcess]},
   {:my_process_6, #PID<0.262.0>, :worker, [MyProcess]},
   {:my_process_5, #PID<0.259.0>, :worker, [MyProcess]},
   {:my_process_4, #PID<0.256.0>, :worker, [MyProcess]},
   {:my_process_3, #PID<0.253.0>, :worker, [MyProcess]},
   {:my_process_2, #PID<0.250.0>, :worker, [MyProcess]}
 ]}

  
    
  
  Querying a specific child


To quickly get the PID of the running process by it's child_id we can use the
ProcessHub.get_pid/2 function which returns the process identifier.
iex> ProcessHub.get_pid(:my_hub, :my_process_10)
#PID<0.228.0>
Now when utilizing replication strategies we should default to another function ProcessHub.get_pids/2
which returns a list of PIDs for the given child_id.
iex> ProcessHub.get_pids(:my_hub, :my_process_10)
[#PID<0.228.0>]
We may also query the whole child info by using the ProcessHub.which_children/2 function.
This function also gives us the child_spec.
iex> ProcessHub.child_lookup(:my_hub, :my_process_10)
{
    %{id: :my_process_10, start: {MyProcess, :start_link, [nil]}},
    [one@anuar: #PID<0.228.0>]
}

  
    
  
  Process metadata


We can store metadata in the registry by starting the process with :metadata option.
# First we need to start the process with metadata
iex> ProcessHub.start_child(
  :my_hub, 
  %{id: :my_process_1, start: {MyProcess, :start_link, [nil]}}, 
  [metadata: %{some_info: "my data"}, awaitable: true]
) |> ProcessHub.Future.await()
%ProcessHub.StartResult{
  status: :ok,
  started: [my_process_1: ["node1@127.0.0.1": #PID<0.293.0>]],
  errors: [],
  rollback: false
}

# Now we can query the process with metadata
iex> ProcessHub.child_lookup(:my_hub, :my_process_1, [with_metadata: true])
{
  %{id: :my_process_1, start: {MyProcess, :start_link, [nil]}},
  [one@anuar: #PID<0.228.0>],
  %{some_info: "my data"}
}
We can also get all the metadata stored for all the processes in the registry by using
ProcessHub.registry_dump/1 function.
iex> ProcessHub.registry_dump(:my_hub)
%{
  my_process_1: {%{id: :my_process_1, start: {MyProcess, :start_link, [nil]}},
   [one@anuar: #PID<0.228.0>], %{some_info: "my data"}},
  my_process_2: {%{id: :my_process_2, start: {MyProcess, :start_link, [nil]}},
   [one@anuar: #PID<0.250.0>], %{}},
  my_process_3: {%{id: :my_process_3, start: {MyProcess, :start_link, [nil]}},
   [one@anuar: #PID<0.253.0>], %{}}
}
ProcessHub is also internally using :metadata to tag processes which is then used
to query the processes by tags.
iex> ProcessHub.tag_query(:my_hub, "MY_TAG")
[
  my_process_1: {%{id: :my_process_1, start: {MyProcess, :start_link, [nil]}},
   [one@anuar: #PID<0.228.0>]},
  my_process_2: {%{id: :my_process_2, start: {MyProcess, :start_link, [nil]}},
   [one@anuar: #PID<0.250.0>]}
]


  

    
Starting and stopping processes
    

The ProcessHub library provides a way to start and stop processes dynamically. 
The processes can be started under a hub and can be stopped at any time.
ProcessHub can be used with any process as long as it returns {:ok, pid} when started. We recommend using GenServer for the processes as it provides a lot of features out of the box.

  
    
  
  Process identification


The :id of the child specification is used to uniquely identify the process in the hub and can be used to stop the process or query its pid.
The :id has to be either an atom or a string. For systems that create lots of dynamic processes, it is recommended to use strings as :id because atoms are not garbage collected. 
The :id has to be unique within the hub. If a process with the same :id is started again, it may result in unexpected behavior unless the uniqueness is checked before starting the process using the :check_existing option.

  
    
  
  Starting processes


Processes can be started dynamically using the ProcessHub.start_children/3 or ProcessHub.start_child/3 functions.
The latter is a convenience function to start a single process and wraps the former.
Some examples of starting processes are:

  
    
  
  Starting multiple processes


iex> ProcessHub.start_children(:my_hub, [
  %{id: "process1", start: {MyProcess, :start_link, [nil]}},
  %{id: "process2", start: {MyProcess, :start_link, [nil]}}
])
{:ok, :start_initiated}

  
    
  
  Starting a single process


child_spec = %{id: "process_id", start: {MyProcess, :start_link, [nil]}}
ProcessHub.start_child(:my_hub, child_spec)
{:ok, :start_initiated}

  
    
  
  Starting processes statically


Processes can also be started statically under the supervision tree. This is useful when the processes are known at compile time.
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    child_specs = [
        %{
            id: :my_process_1,
            start: {MyProcess, :start_link, [nil]}
        },
        %{
            id: :my_process_2,
            start: {MyProcess, :start_link, [nil]}
        }
    ]

    children = [
      ProcessHub.child_spec(%ProcessHub{
        hub_id: :my_hub,
        child_specs: child_specs
      })
    ]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end
end
See ProcessHub.init_opts/0 for more information on the options that can be passed to the start functions.

  
    
  
  Stopping processes


Processes can be stopped dynamically using the ProcessHub.stop_child/3 or ProcessHub.stop_children/3 functions.
Some examples of stopping processes are:
ProcessHub.stop_child(:my_hub, "child_id")
{:ok, :stop_initiated}
ProcessHub.stop_children(:my_hub, ["child_id1", "child_id2"])
{:ok, :stop_initiated}
See ProcessHub.stop_opts/0 for more information on the options that can be passed to the stop functions.

  
    
  
  Self-stopping processes


Processes can also stop themselves by sending exit signals to the linked processes (supervisor).
This is useful when the process has to stop itself based on some condition.
For processes to be eligible for self-stopping, they have to be started with a :restart option set to :transient.
If the :restart option is anything other than :transient and the stop code is not either shutdown or normal, the supervisor will try to restart the process when it stops.
This will also update the registry accordingly for all the nodes in the cluster. Keep in mind that the operation is asynchronous.

  
    
  
  Example self-stopping GenServer process


child_spec1 = %{
      id: :self_shutdown_1,
      start: {Test.Helper.TestServer, :start_link, [%{name: :self_shutdown_1}]},
      restart: :transient # Set the restart option to transient
    }

ProcessHub.start_child(:my_hub, child_spec)

# In the process module
defmodule MyProcess do
  use GenServer

  def start_link(_) do
    GenServer.start_link(__MODULE__, nil)
  end

  def init(_) do
    {:ok, nil}
  end

  def handle_info(msg, state) do
    {:stop, :normal, state} # Return the stop tuple
  end
end

  
    
  
  Asynchronous operations


All process starting and stopping operations are asynchronous by default and can be made synchronous by passing the awaitable: true option. 
By doing so, the operation returns a ProcessHub.Future struct that can be awaited.

  
    
  
  Example of starting a process and awaiting the result


child_spec = %{
  id: "my_process_1",
  start: {MyProcess, :start_link, [nil]}
}
ProcessHub.start_children(:my_hub, [child_spec], awaitable: true) 
|> ProcessHub.Future.await()
%ProcessHub.StartResult{
  status: :ok,
  started: [{"my_process_1", ["node1@127.0.0.1": #PID<0.231.0>]}],
  errors: [],
  rollback: false
}

  
    
  
  Example of stopping a process and awaiting the result


ProcessHub.stop_child(:my_hub, "my_process_1", awaitable: true)
|> ProcessHub.Future.await()
%ProcessHub.StopResult{
  status: :ok,
  stopped: [{"my_process_1", [:"node1@127.0.0.1"]}],
  errors: []
}

  
    
  
  Formatting and extracting information from the results


When using awaitable operations, ProcessHub returns ProcessHub.StartResult and ProcessHub.StopResult structs that contain detailed information about the operation. These structs provide both formatting functions and convenient accessor functions to extract specific information.

  
    
  
  StartResult API Functions


The ProcessHub.StartResult module provides several functions to extract information from start operation results:
Basic Information Extraction
# Start processes and get the result
result = ProcessHub.start_children(:my_hub, [child_spec1, child_spec2], awaitable: true)
|> ProcessHub.Future.await()

# Check the overall status
ProcessHub.StartResult.status(result)
:ok

# Get all child IDs that were processed
ProcessHub.StartResult.cids(result)
["child1", "child2"]

# Get all errors that occurred
ProcessHub.StartResult.errors(result)
[]
Process and Node Information
# Get the first PID from the first started process
ProcessHub.StartResult.pid(result)
#PID<0.123.0>

# Get all PIDs from all started processes
ProcessHub.StartResult.pids(result)
[#PID<0.123.0>, #PID<0.124.0>]

# Get the first node from the first started process
ProcessHub.StartResult.node(result)
:"node1@127.0.0.1"

# Get all unique nodes where processes were started
ProcessHub.StartResult.nodes(result)
[:"node1@127.0.0.1", :"node2@127.0.0.1"]

# Get the first started child entry (full details)
ProcessHub.StartResult.first(result)
{"child1", [{:"node1@127.0.0.1", #PID<0.123.0>}]}

  
    
  
  StopResult API Functions


The ProcessHub.StopResult module provides similar functions for stop operation results:
# Stop processes and get the result
result = ProcessHub.stop_children(:my_hub, ["child1", "child2"], awaitable: true)
|> ProcessHub.Future.await()

# Check the overall status
ProcessHub.StopResult.status(result)
:ok

# Get all child IDs that were processed
ProcessHub.StopResult.cids(result)
["child1", "child2"]

# Get all errors that occurred
ProcessHub.StopResult.errors(result)
[]

# Get all unique nodes where processes were stopped
ProcessHub.StopResult.nodes(result)
[:"node1@127.0.0.1", :"node2@127.0.0.1"]

# Get the first stopped child entry
ProcessHub.StopResult.first(result)
{"child1", [:"node1@127.0.0.1"]}

  
    
  
  Chaining Operations


You can chain ProcessHub operations with await and result processing in a single pipeline:
# Chain start operation with immediate formatting
{:ok, started_processes} = ProcessHub.start_children(:my_hub, [child_spec1, child_spec2], awaitable: true)
|> ProcessHub.Future.await()
|> ProcessHub.StartResult.format()

# Chain start operation with status checking
:ok = ProcessHub.start_child(:my_hub, child_spec, awaitable: true)
|> ProcessHub.Future.await()
|> ProcessHub.StartResult.status()

# Chain stop operation with child ID extraction
stopped_ids = ProcessHub.stop_children(:my_hub, ["child1", "child2"], awaitable: true)
|> ProcessHub.Future.await()
|> ProcessHub.StopResult.cids()
# ["child1", "child2"]

# Chain start operation with PID extraction
pids = ProcessHub.start_children(:my_hub, [child_spec1, child_spec2], awaitable: true)
|> ProcessHub.Future.await()
|> ProcessHub.StartResult.pids()
# [#PID<0.123.0>, #PID<0.124.0>]

  
    
  
  Pattern Matching on Result Structs


Both StartResult and StopResult are regular Elixir structs that can be pattern matched directly:
# Pattern match on successful start
result = ProcessHub.start_children(:my_hub, [child_spec1, child_spec2], awaitable: true)
|> ProcessHub.Future.await()

case result do
  %ProcessHub.StartResult{status: :ok, started: started} ->
    IO.puts("Successfully started: #{inspect(started)}")
    
  %ProcessHub.StartResult{status: :error, errors: errors, started: started} ->
    IO.puts("Some failures: #{inspect(errors)}")
    IO.puts("Some successes: #{inspect(started)}")
end

  
    
  
  Format Functions


Both result types provide format/1 functions that return tuple formats:
ProcessHub.StartResult.format(result)
{:ok, [{"child1", [{:"node1@127.0.0.1", #PID<0.123.0>}]}]}

# For error cases with rollback
{:error, {[{"child1", :timeout}], [{"child2", [{:"node1@127.0.0.1", #PID<0.124.0>}]}]}, :rollback}

ProcessHub.StopResult.format(result)  
{:ok, [{"child1", [:"node1@127.0.0.1"]}]}

# For error cases
{:error, {[{"child1", :"node1@127.0.0.1", :not_found}], []}}


  

    
Configuring the hub
    


  
    
  
  Configurable strategies


ProcessHub comes with 12 different strategies that can be used to configure the hub.
All strategies are Elixir structs that implement their own base protocol.
In fact, you can define your own strategies by implementing the base protocols.
When configuring the hub, you can pass the strategies as part of the %ProcessHub{} struct.
Look at the documentation for each strategy for more information on how to configure them.
An example can be seen below.
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    children = [process_hub()]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end

  defp process_hub() do
    {ProcessHub, %ProcessHub{
      hub_id: :my_hub,
      # Configure the redundancy strategy.
      redundancy_strategy: %ProcessHub.Strategy.Redundancy.Replication{
        replication_factor: 2,
        replication_model: :active_passive,
        redundancy_signal: :none
      },
      # Configure the migration strategy.
      migration_strategy: %ProcessHub.Strategy.Migration.HotSwap{
        retention: 2000,
        handover: true
      },
      # Configure the synchronization strategy.
      synchronization_strategy: %ProcessHub.Strategy.Synchronization.PubSub{
        sync_interval: 10000
      },
      # Configure the partition tolerance strategy.
      partition_tolerance_strategy: %ProcessHub.Strategy.PartitionTolerance.StaticQuorum{
        quorum_size: 3
      },
      # Configure the distribution strategy.
      distribution_strategy: %ProcessHub.Strategy.Distribution.ConsistentHashing{}
    }}
  end
end
Each strategy module has additional documentation on how to configure it.

  
    
  
  Redundancy Strategy


ProcessHub.Strategy.Redundancy.Base - defines the base protocol for redundancy strategies.
This strategy is used to define how many replicas of a process should be started
across the cluster. Starting multiple instances of a process across the cluster is useful
for redundancy and fault tolerance.
Available strategies are:
	ProcessHub.Strategy.Redundancy.Singularity - only 1 process per child_id without
any replicas. This is also the default strategy and contains no special configuration options.
	ProcessHub.Strategy.Redundancy.Replication - starts multiple replicas of a process
across the cluster. The number of replicas is defined by the :replication_factor
option. This strategy also supports :active_active and :active_passive modes.
Meaning we may have one active process and the rest are passive.
The mode is defined by the :replication_model option.
This information will be passed to the started process.
The default mode is :active_active, meaning all processes are equal and considered active.


  
    
  
  Migration Strategy


ProcessHub.Strategy.Migration.Base - defines the base protocol for migration strategies.
This strategy is used to define how the processes are migrated when a node joins or leaves the cluster.
Migration is the process of moving processes from one node to another.
One of the reasons why migration happens is when a node leaves the cluster.
When a node leaves the cluster, it is possible that some processes are still running on that
node, so these need to be migrated to another node. Also, when a new node joins the
cluster, other nodes may migrate some processes over to the new node.
At the moment, there are 2 migration strategies available:
	ProcessHub.Strategy.Migration.ColdSwap - migrate processes by stopping the process
on the old node before starting it on the new node. This is the default strategy and defines no special configuration options.
	ProcessHub.Strategy.Migration.HotSwap - this strategy is used to migrate processes
by starting the process on the new node before stopping it on the old node.
This strategy is useful when we want to avoid any downtime. This strategy is also
useful when the process is stateful, and we want to avoid any data loss by handing over
the state from the old process to the new process. See the module for handover examples.


  
    
  
  Synchronization Strategy


ProcessHub.Strategy.Synchronization.Base - defines the base protocol for synchronization
strategies which define the method that is used to synchronize the process registry.
Available strategies are:
	ProcessHub.Strategy.Synchronization.PubSub - uses a publish/subscribe model to synchronize
the process registry. Each node in the cluster will subscribe to a topic and publish any changes to the topic.
These changes could be events such as adding or removing a process from the registry.
This is the default and recommended synchronization strategy for most users.
	ProcessHub.Strategy.Synchronization.Gossip - uses a gossip protocol to synchronize the
process registry. Using this strategy is recommended when the number of nodes in the cluster
is large. The Gossip strategy selects a predefined number of nodes to gossip with and
exchange information about the process registry.
These selected nodes will choose other nodes to gossip with and so on until all nodes in the
cluster are synchronized. It has higher latency than the PubSub strategy specially when the cluster
is rather small.


  
    
  
  Partition Tolerance Strategy


ProcessHub.Strategy.PartitionTolerance.Base - defines the base protocol for partition tolerance
strategies which define the method that is used to handle network partitions.
Available strategies are:
	ProcessHub.Strategy.PartitionTolerance.Divergence - this strategy is used to handle network
partitions by diverging the cluster into multiple subclusters. Each subcluster will have its
own hub and will be considered as a separate cluster. This strategy is the default strategy.
When the network partition is healed, the subclusters will merge back into a single cluster.
	ProcessHub.Strategy.PartitionTolerance.StaticQuorum - this strategy is used to handle network
partitions by using a static quorum. The quorum size is defined by the :quorum_size option.
When a partition happens, the ProcessHub.DistributedSupervisor process will terminate
along with its children. This strategy is useful when the number of nodes in the cluster is
known and rather fixed.
	ProcessHub.Strategy.PartitionTolerance.DynamicQuorum - this strategy is used to handle
network partitions by using a dynamic quorum. The quorum size is defined by the :quorum_size
option and :threshold_time option. The system automatically over time adapts to the number of
nodes in the cluster. When a partition happens, the ProcessHub.DistributedSupervisor process
will terminate along with its children.Using DynamicQuorum Strategy
When scaling down too many nodes at once, the system may consider itself to be in a
network partition. Read the documentation for the
ProcessHub.Strategy.PartitionTolerance.DynamicQuorum strategy for more information.




  
    
  
  Distribution Strategy:


ProcessHub.Strategy.Distribution.Base - defines the base protocol for distribution
strategies which define the method to determine the node/process mapping.
ProcessHub currently supports 3 distribution strategies:
	ProcessHub.Strategy.Distribution.ConsistentHashing - uses a consistent hash ring to determine
the node/process mapping. This strategy is the default strategy and operates in a decentralized manner.
Each node is responsible for a range of hash values, and process IDs are hashed to determine their
node assignment. When the cluster topology changes, only affected processes are redistributed,
minimizing unnecessary migrations. Supports process replication and requires no configuration options.

	ProcessHub.Strategy.Distribution.Guided - uses a predefined mapping to determine the
node/process mapping. This strategy is useful when we want to control the mapping of
processes to nodes. You must explicitly specify which nodes each child process should run on
during startup. Processes are bound to their assigned nodes and do not support migration.
This is ideal for scenarios requiring specific node placement based on hardware requirements
or geographic distribution.

	ProcessHub.Strategy.Distribution.CentralizedLoadBalancer - uses a leader-based approach where
a single leader node collects performance metrics from all nodes and makes distribution decisions
based on real-time load data. The leader is determined by highest uptime and is not configurable.
Experimental Strategy
The CentralizedLoadBalancer strategy is currently experimental and should not be used
in production environments.

Configuration options:
	:max_history_size (default: 30) - Maximum historical load scores to maintain
	:weight_decay_factor (default: 0.9) - Exponential decay factor for historical scores
	:push_interval (default: 10,000) - Metrics collection interval in milliseconds
	:nodeup_redistribution (default: false) - Process redistribution on node join (should not be changed)





  

    
Process state handover
    

In this guide we will show a simple example of how to handover the state of a process from one node to another.
This scenario is happening automatically when the cluster has received new node event,
meaning that the new node has joined the cluster or left.
Suppose we have the following GenServer process:
defmodule MyProcess do
  use GenServer

  def start_link(state) do
    GenServer.start_link(__MODULE__, state)
  end

  def init(state) do
    {:ok, state}
  end
end
In order to handover the state of the process from one node to another, we need to do the following:

  
    
  
  1. Configure the hub


Configure the ProcessHub to use the Hotswap strategy.
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    children = [process_hub()]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end

  defp process_hub() do
    {ProcessHub, %ProcessHub{
      hub_id: :my_hub,
      migration_strategy: %ProcessHub.Strategy.Migration.HotSwap{
        retention: 3000,
        handover: true
      },
    }}
  end
end
Pay attention to the retention and handover options. The retention option is the max time in milliseconds that the old process will be kept alive when migrating. The handover option is a boolean that indicates whether the process state should be handed over to another node when the process is going to be stopped.
The HotSwap module also supports a option called :confirm_handover which works togather with the :handover option. The :confirm_handover option is a boolean that indicates whether the migration handler should wait for the confirmation message from the new process about the handover. If this option is set to true, the :children_migrated_hook hook will be called once the handover is confirmed, making it possible to react to the handover process.
The :confirm_handover option is triggered on node join or leave events but not on node shutdown events.
Retention option
The retention option does not mean that the old process will be kept that long alive.
It is used to limit the time the old processes can stay alive in the handover process
before they are killed.


  
    
  
  2. Implement the neccessary callbacks


In order to handover the state our MyProcess have to implement the ProcessHub.Strategy.Migration.Handover behaviour or define the necessary callbacks.
This can be achieved by using the ProcessHub.Strategy.Migration.HotSwap module that provides the necessary callbacks and the default implementation of the prepare_handover_state/1 and alter_handover_state/2 callbacks
defmodule MyProcess do
  use GenServer
  use ProcessHub.Strategy.Migration.HotSwap # Provides the necessary callbacks
end
Users are free to implement the alter_handover_state/2 callback in order to modify the state of the process before it is handed over to another node.
defmodule MyProcess do
  use GenServer
  use ProcessHub.Strategy.Migration.HotSwap # Provides the necessary callbacks

  # This is called on the process that is going to pass the state to another
  # newly spawned process. The returned value of this function will be passed to the
  # `alter_handover_state/2` function on the process that is going to accept the new state.
  @impl true
  def prepare_handover_state(state) do
    state
  end

  # This is called on the process that is going to accept the new handover state.
  # The process must return the new state that will be used by the process.
  @impl true
  def alter_handover_state(current_state, handover_state) do
    new_value = Map.get(handover_state, :some_key)

    # Do something with the state before returning it
    new_state = Map.put(current_state, :some_key, new_value)

    # Do something else..

    # Return the new state
    new_state
  end
end

  
    
  
  3. That's it!


Now the ProcessHub will take care of the rest. When the process is going to be
redirected to another node, the state of the process will be handed over to the new process on
the new node.

  
    
  
  Handover on node leave


The hub takes care of the handover process when a node joins the cluster and processes
migrate to the new node. 
The same can be said when a node leaves the cluster, meaning that the process states on the 
leaving node will be handed over to the remaining nodes but only if the node is shut down gracefully.
This leaves the hub with the time to handover the process states to the remaining nodes.
It is advised to scale nodes down one at a time to avoid any data loss(30 second - 10 minute intervals).

  
    
  
  More info can be found in the Hotswap module


Check out the Hotswap module for more information.


  

    
Hooks
    

Hooks are used to trigger events on specific actions. Hooks can be registered by passing the
handlers to the :hooks option of the ProcessHub.t/0 configuration struct or by inserting them
dynamically using the ProcessHub.Service.HookManager module.
ProcessHub heavily uses hooks internally in the integration tests and strategy implementations.
Hooks have to be in the format of an mfa tuple. Basically, they are functions that will be called
when the hook is triggered.
It is possible to register a hook handler with a wildcard argument :_, which will be replaced with the hook data when the hook is dispatched.

  
    
  
  Hook registration examples



  
    
  
  Static hook registration


Static hook registration occurs when all hook handlers are known at the time of hub initialization
using the ProcessHub.t/0 configuration struct.
Example:
# Register a hook handler for the `:pre_cluster_join_hook` event with a wildcard argument.
defmodule MyApp.Application do
  alias ProcessHub.Service.HookManager
  alias alias ProcessHub.Constant.Hook

  use Application

  def start(_type, _args) do
    children = [
      ProcessHub.child_spec(%ProcessHub{
        hub_id: :my_hub,
        hooks: %{
          Hook.pre_cluster_join() => [
            %HookManager{
              id: :hook_id_1,
              m: MyModule,
              f: :my_function,
              a: [:some_data, :_]
            },
            %HookManager{
              id: :hook_id_2,
              m: MyModule2,
              f: :my_function2,
              a: [:some_data]
            }
          ],
          Hook.post_cluster_join() => [
            %HookManager{
              id: :hook_id_3,
              m: MyModule,
              f: :my_function,
              a: [:some_data, :_]
            }
          ]
        }
      })
    ]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end
end

  
    
  
  Dynamic hook registration


Dynamic hook registration is useful when modifying hook handlers at runtime. It allows us to add new handlers or remove existing ones.
Example:
alias ProcessHub.Service.HookManager
alias ProcessHub.Constant.Hook

# Register a hook handler for the `:cluster_join` event with a wildcard argument.
ProcessHub.register_hook_handlers(
  :my_hub, 
  Hook.post_cluster_join(),
  %HookManager{
    id: :hook_id_30, 
    m: MyModule,
    f: :my_function,
    a: [:some_data, :_]
  }
)

  
    
  
  Hook handler examples


# The hook handler should be in the following format:
defmodule MyModule do
  def my_function(:some_data, dynamic_hook_data) do
    # Do something with the data.
  end
end

  
    
  
  Using hooks to react and handle asynchronous events


Sometimes when starting a child process, it is necessary to know when the child process has been
started and is ready to receive messages. We could use the :awaitable option with the ProcessHub.Future.await/1 but that would
block the process that is starting the child process. Instead, we can use hooks to react to the
:pre_children_start event asynchronously.
Example:
# Register hook handler
alias ProcessHub.Service.HookManager
alias ProcessHub.Constant.Hook
...
hub_id = :my_hub

hook_handler1 = %HookManager{
  id: :handler_id_1,
  m: MyModule,
  f: :my_first_func,
  a: [:_]
}

hook_handler2 = %HookManager{
  id: :handler_id_2,
  m: MyModule,
  f: :my_second_func,
  a: [:hub_id, :_]
}

# We're triggering the hook when child process has been registered under the process registry.
ProcessHub.register_hook_handlers(:my_hub, Hook.registry_pid_inserted(), [hook_handler1, hook_handler2])
...

# The hook handlers
defmodule MyModule do
  def my_first_func(hub_id, {child_id, node_pids}) do
    # Send message to all started child processes.
    Enum.each(node_pids, fn {_node, pid} ->
      Process.send(pid, :hello_world, [])
    end)
  end

  def my_second_func(hub_id, hook_data) do
    ...
  end
end

  
    
  
  Internally registered hook handlers


ProcessHub is using hooks internally to depending on the active configuration.
These hook handlers should never be removed or modified.
The list of internally registered hooks ids are: 
	:ch_join
	:ch_leave
	:ch_shutdown
	:mhs_shutdown
	:mhs_process_startups
	:rr_post_start
	:rr_post_stop
	:rr_post_update
	:dg_pre_start_handler


  
    
  
  Hooks in custom strategy implementations


Hooks can be used in custom strategy implementations to trigger events on specific actions and
extend the functionality of the strategy.
Here's an example of ProcessHub using hooks internally in the 
ProcessHub.Strategy.Distribution.ConsistentHashing strategy implementation:
alias ProcessHub.Constant.StorageKey
alias ProcessHub.Service.Storage
alias ProcessHub.Service.HookManager

# Register hook handlers to update the hash ring when node joins or leaves the cluster.
def init(_strategy, hub) do
  hub_nodes = Storage.get(hub.storage.misc, :hub_nodes)

  Storage.insert(hub.storage.misc, StorageKey.hr, Ring.create_ring(hub_nodes))

  join_handler = %HookManager{
    id: :ch_join,
    m: ProcessHub.Strategy.Distribution.ConsistentHashing,
    f: :handle_node_join,
    a: [hub, :_]
  }

  leave_handler = %HookManager{
    id: :ch_join,
    m: ProcessHub.Strategy.Distribution.ConsistentHashing,
    f: :handle_node_leave,
    a: [hub, :_]
  }

  shutdown_handler = %HookManager{
    id: :ch_shutdown,
    m: ProcessHub.Strategy.Distribution.ConsistentHashing,
    f: :handle_shutdown,
    a: [hub],
    p: 100
  }

  HookManager.register_handler(hub.storage.hook, Hook.pre_cluster_join(), join_handler)
  HookManager.register_handler(hub.storage.hook, Hook.pre_cluster_leave(), leave_handler)
  HookManager.register_handler(hub.storage.hook, Hook.coordinator_shutdown(), shutdown_handler)
end


# Handler functions that will be called when the hooks are triggered.
def handle_node_join(hub, node) do
  hash_ring = Ring.get_ring(hub.storage.misc) |> Ring.add_node(node)

  Storage.insert(hub.storage.misc, StorageKey.hr, hash_ring)
end

def handle_node_leave(hub, node) do
  hash_ring = Ring.get_ring(hub.storage.misc) |> Ring.remove_node(node)

  Storage.insert(hub.storage.misc, StorageKey.hr, hash_ring)
end

  
    
  
  Hook handler priority


ProcessHub by default does not guarantee the order of hook handler execution unless 
the hook handler specifies a priority level. The priority level is an integer value that
determines the order of execution of the hook handlers. The higher the priority level, the
earlier the hook handler will be executed. By default all hooks have a priority level of 0.
We define the priority level of a hook handler by setting the :p field of the ProcessHub.Service.HookManager.t/0
Here an example of how to set the priority level of a hook handler to 15:
alias ProcessHub.Service.HookManager
alias ProcessHub.Constant.Hook
...
hook_handler = %HookManager{
  id: :handler_id_1,
  m: MyModule,
  f: :handle_child_start,
  a: [],
  p: 15 # here we set the priority level to 15.
}

ProcessHub.register_hook_handlers(:my_hub, Hook.pre_cluster_leave(), hook_handler)

  
    
  
  Hook handler arguments


The hook handler expects a arguments which is a list of values. 
These values will then be passed to the hook handler function in the same order.
Some hooks dispatchers will pass additional data to the hook handler function which
then can be accessed using the wildcard argument in the handler function.
These are called wildcard arguments and are represented by the :_ symbol.
Example:
alias ProcessHub.Service.HookManager
alias ProcessHub.Constant.Hook
...
hook_handler = %HookManager{
  id: :handler_id_1,
  m: MyModule,
  f: :handle_child_start,
  a: [:_, :some, :something_else] # here the first argument will be the wildcard argument.
}

ProcessHub.register_hook_handlers(:my_hub, Hook.pre_cluster_join(), hook_handler)

# Here the handler function receives the wildcard argument as the first argument.
defmodule MyModule do
  def handle_child_start(wild_card_value, :some, :something_else) do
    # Because we used the `:pre_cluster_join_hook` hook event the wild_card_value will be a `node()`
  end
end
You can familiarize yourself with the available hooks and their data in the last section of this guide.

  
    
  
  Listing hook handlers


You can list all registered hook handlers for a specific hook event by calling 
the registered_handlers/2 function from the ProcessHub.Service.HookManager module.
Example:
iex> alias ProcessHub.Service.HookManager
iex> alias ProcessHub.Constant.Hook 
iex> HookManager.registered_handlers :my_hub, Hook.registry_pid_inserted()  
[
  %ProcessHub.Service.HookManager{
    id: :handler_id_2,
    m: MyModule,
    f: :my_second_func,
    a: [:hub_id, :_],
    p: 0
  }
]

  
    
  
  Removing hook handlers


You can remove a specific hook handler by calling the cancel_handler/3 function.
Example:
iex> alias ProcessHub.Service.HookManager
iex> alias ProcessHub.Constant.Hook 
iex> HookManager.cancel_handler(:my_hub, Hook.registry_pid_inserted(), :handler_id_1)
:ok
It's also possible to dispatch hooks manually by calling the dispatch_hook/3 or
dispatch_hooks/3 functions from the ProcessHub.Service.HookManager module.
Although, it's not recommended to dispatch hooks manually as it can lead to unexpected behavior.

  
    
  
  Available hooks


	Event Key	Trigger	Data
	:pre_cluster_join_hook	Node joins hub cluster, not handled	node()
	:post_cluster_join_hook	Node joins hub cluster, handled	node()
	:pre_cluster_leave_hook	Node leaves hub cluster, not handled	node()
	:post_cluster_leave_hook	Node leaves hub cluster, handled	node()
	:registry_pid_insert_hook	Process registered	{child_id(), [{node(), pid()}]}
	:registry_pid_remove_hook	Process unregistered	child_id()
	:children_migrated_hook	Processes migrated	{node(), [child_spec()], map()}
	:priority_state_updated_hook	Priority state updated	{priority_level(), list()}
	:pre_nodes_redistribution_hook	Nodes redistribution start	{:nodeup or :nodedown, node()}
	:post_nodes_redistribution_hook	Nodes redistribution end	{:nodeup or :nodedown, node()}
	:forwarded_migration_hook	Migration was forwarded and handled	[{node(), [children_start_data()]}]
	:pre_children_start_hook	Before child process is started	ProcessHub.Handler.ChildrenAdd.StartHandle.t()
	:post_children_start_hook	After child process is started	[{child_id(), result(), pid(), [node()]}]
	:pre_children_redistribution_hook	Before redistribution of children is called	[{child_id(), node(), {:pid, pid()}}]
	:coordinator_shutdown_hook	Coordinator shutdown	any()
	:process_startups_hook	Processes have been started	[%{cid: child_id, pid: pid()}]
	:child_process_pid_update_hook	Child process restarted with new pid	{node(), child_id()}


  
    
  
  Alter hooks


Alter hooks differ from the other hooks in that they are used to modify the hook data
before it is passed to the function that is using that data. This allows us to alter the data
instead of just reacting to it and cause side effects as in the regular hooks.
Registering an alter hook is exactly the same as registering a normal hook.
Alter hooks must return the modified data in the same format as the original data.
⚠️ Warning
Alter hooks should be used with caution as they can lead to unexpected behavior if not used properly.
For example, if you modify the hook data in an alter hook, it will be passed to the next hook handler
and can cause unexpected behavior if the next hook handler or the code is not expecting the modified data.
Never change the underlying data structure of the hook data such as from map to list etc.


  
    
  
  List of alter hooks


	Event Key	Trigger	Data
	:child_data_alter_hook	Before supervisor starts the child process	map() including child spec, metadata and nodes

See ProcessHub.Constant.Hook module for more information on available hooks.


  

    
Manual distribution
    


  
    
  
  Introduction


By default ProcessHub uses the ProcessHub.Strategy.Distribution.ConsistentHashing strategy to distribute processes.
This is automatic and the ProcessHub will decide which node to start the process on.
This strategy is useful for most use cases, but sometimes you may want to distribute processes manually. This guide will show you how to do that.
By manually distributing processes, you can control which processes are started on which nodes. 
This can be useful when you want to distribute processes based on specific criteria, such as the node's hardware capabilities or the process's workload.
Let's dive in a see some examples.

  
    
  
  Example GenServer


As always our processes should have some template to follow. Here is a simple GenServer process that we will use in our examples:
defmodule MyProcess do
  use GenServer

  def start_link(state) do
    GenServer.start_link(__MODULE__, state)
  end

  def init(state) do
    {:ok, state}
  end
end

  
    
  
  The configuration


First, let's configure the ProcessHub to use the ProcessHub.Strategy.Distribution.Guided strategy.
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    children = [process_hub()]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end

  defp process_hub() do
    {ProcessHub, %ProcessHub{
      hub_id: :my_hub,
      # We have to swap the default distribution strategy with the guided one.
      distribution_strategy: %ProcessHub.Strategy.Distribution.Guided{}
    }}
  end
end

  
    
  
  Start 2 nodes


Now let's start 2 nodes and connect them to each other.
iex --name node1@127.0.0.1 --cookie mycookie -S mix

iex --name node2@127.0.0.1 --cookie mycookie -S mix

From node 1, connect to node 2(or vice versa):
Node.connect(:"node2@127.0.0.1")

  
    
  
  Start distributed processes.


Now let's start a process on node 1 and guide it to node 2.
iex> node1_child = %{id: :node1_child, start: {MyProcess, :start_link, [nil]}}
iex> node2_child = %{id: :node2_child, start: {MyProcess, :start_link, [nil]}}
iex> ProcessHub.start_children(:my_hub, [node1_child, node2_child], [
...>    child_mapping: %{
...>        node1_child: [:"node1@127.0.0.1"],
...>        node2_child: [:"node2@127.0.0.1"],
...>    }
...> ])
In the example above, we started 2 processes, node1_child and node2_child. We also specified that node1_child should be started on node 1 and node2_child should be started on node 2.
We do this by adding the child_mapping option to the ProcessHub.start_children/3 function. The child_mapping option is a map that specifies which node to start the process on.
Each key in the map is the id of the process and the value is a list of nodes to start the process on.
Why use lists to specify the node?
The reason we use lists for the child_mapping is that we can specify multiple nodes for a 
process if we want to replicate the process on multiple nodes. 
This requires the ProcessHub.Strategy.Redundancy.Replication strategy to be used. Which we will cover in another guide.
For example: node1_child: [:"node1@127.0.0.1", :"node2@127.0.0.1"]


  
    
  
  Checking the registered processes


Now let's check the registered processes on each node.
iex> ProcessHub.process_list(:my_hub, :global)
[
  node1_child: ["node1@127.0.0.1": #PID<0.280.0>],
  node2_child: ["node2@127.0.0.1": #PID<22474.276.0>]
]
We can see that the node1_child process is registered on node 1 and the node2_child process is registered on node 2.
The child_mapping is mandatory
If we decide to use the ProcessHub.Strategy.Distribution.Guided strategy, we have to specify the child_mapping option when starting the processes. Otherwise {:error, :missing_child_mapping} will be returned when starting the processes.



  

    
Replicating processes
    

In a distributed system, it can be important to replicate processes within a cluster of nodes. This can lead to better resilience and reliability of the system.
When one node unexpectedly fails, the replicated processes on other nodes can take over the work of the failed node immediately.
In this guide, we will show a simple example of how to replicate processes and prioritize them.

  
    
  
  Prerequisites


As always we should have some process template to follow. You can re use the MyProcess GenServer from the Manual Distribution guide.

  
    
  
  The configuration


First, let's configure the ProcessHub to use the ProcessHub.Strategy.Redundancy.Replication strategy.
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    children = [process_hub()]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end

  defp process_hub() do
    {ProcessHub, %ProcessHub{
      hub_id: :my_hub,
      redundancy_strategy: %ProcessHub.Strategy.Redundancy.Replication{    
        replication_factor: 2,
        replication_model: :active_passive,
        redundancy_signal: :all
      },
    }}
  end
end
Let's explain the used options:
	replication_factor: The number of replicas that will be created for each process. Meaning if the value is 2, then the process will be replicated on 2 nodes.
	replication_model: This is optional and can be either :active_passive or :active_active. The :active_passive model means that only one of the replicas will be marked as active and the others will be passive. We will get into more details about this later.
	redundancy_signal: This is optional and can be used to toggle on/off the message sent
to the replicas to notify them about their current state (:passive or :active). This all will be explained in the next sections.


  
    
  
  Start 2 nodes


Now let's start two nodes and connect them to each other. 
You can refer to the Manual Distribution guide for an example.

  
    
  
  Start child


Next let's start a process on one node and then see how it is replicated on the other node.
We will be using the awaitable just for demonstration purposes and see the results printed in the console.
iex> child_spec = %{id: :child1, start: {MyProcess, :start_link, [nil]}}
iex> ProcessHub.start_children(:my_hub, [child_spec], [awaitable: true]) 
iex> |> ProcessHub.Future.await()
iex> |> ProcessHub.StartResult.format()
{:ok,
 [
   child1: [
     "node1@127.0.0.1": #PID<0.264.0>,
     "node2@127.0.0.1": #PID<22914.996.0>
   ]
 ]}
***17:08:08.277 [error] MyProcess #PID<0.264.0> received unexpected message in handle_info/2: {:process_hub, :redundancy_signal, :passive}***
If we examine the result, we can see that the process child1 was started on both nodes, meaning it was replicated.
But what's up with the error message?
Well, the error message is expected. It is a signal sent to the process to notify it that it is either in passive or active mode. This is part of the redundancy_signal option we set in the ProcessHub configuration.
We could have used the redundancy_signal: :none option to disable the signal, but we wanted to show you how it works.
Redundancy Signal
The redundancy_signal option is used to toggle on/off the message sent to the replicas to notify them about their current state (:passive or :active). 
This message needs to be handled by the process.

Here's an example how we handle the signal in the MyProcess. In the following example, we
save the signal in the state and handle incoming messages based on the state.
# The required callback we have to implement, to handle the signal.
def handle_info({:process_hub, :redundancy_signal, mode}, state) do
    # Save the mode in the state for later use.
    {:noreply, %{mode: mode}}
end

# Below are examples of how we can handle the signal in the process.
# These are just examples and you can implement your own logic based on the mode.

# Here we define two different handle_info functions based on the mode.
# The second one handles the request because we're the active process.
# The first one does nothing because we're the passive process.

def handle_info({:handle_request, _request}, %{mode: :passive} = state) do
    # Ignore and don't do anything because we're the passive process.
    {:noreply, state}
end

def handle_info({:handle_request, request}, %{mode: :aactive} = state) do
    # Handle the request because we're the active process.
    handle_request(request)
    {:noreply, state}
end

  
    
  
  Conclusion


The ProcessHub.Strategy.Redundancy.Replication can be used not only for replicating processes but also to manipulate the behavior of the replicas. This can be done by using the :redundancy_signal with :replication_model options.
ProcessHub defines currently two redundancy strategies, You can check these modules for custom configuration options:
	ProcessHub.Strategy.Redundancy.Replication
	ProcessHub.Strategy.Redundancy.Singularity



  

    
Creating custom strategy
    

It's time to extend the functionality of the ProcessHub and we will do this
by creating a custom distribution strategy.

  
    
  
  Prerequisites


As always, we should have some process template to follow. You can reuse the MyProcess GenServer from the Manual Distribution guide.

  
    
  
  Implementation


Our custom distribution strategy will be very simple and probably not very
useful in practice, but it will serve as a good example of how to create one.
This distribution strategy will distribute processes by comparing the node names. We will also take advantage
of the hook system to further customize the behavior of the system.
Let's start by creating a new file called compare.ex. We can place the file anywhere
in the project as long as it is picked up by the compiler.
It is a good practice to add some custom namespacing to the module to avoid conflicts
with other modules. We will use CustomStrategy as the namespace.
defmodule CustomStrategy.Compare do
end
Our custom distribution strategy should define a struct which can contain any 
number of custom options. In our example, we will define a single option called
direction which will determine if the selected node is either the first or the last
node in the list of nodes.
We will also create our implementation of the ProcessHub.Strategy.Distribution.Base protocol.
defmodule CustomStrategy.Compare do
    alias ProcessHub.Strategy.Distribution.Base, as: DistributionStrategy

    # We can define custom options.
    defstruct [
        direction: :asc     
    ]

    defimpl DistributionStrategy, for: CustomStrategy.Compare do
    end
end
The ProcessHub.Strategy.Distribution.Base protocol defines some functions that we need to implement. The first function is the init/2 function which is called very early when
the coordinator is started. This can be a good place to register any custom hook handlers that we need to use. We're going to use this function to register a hook handler that will print the name of
the node that is joining the hub just to demonstrate the hook system.
...
defimpl DistributionStrategy, for: CustomStrategy.Compare do
    alias ProcessHub.Service.HookManager
    alias ProcessHub.Constant.Hook

    @impl true
    def init(_struct, %ProcessHub.Hub{} = hub) do
      handler = %HookManager{
        id: :some_id_here,
        m: CustomStrategy.Compare,
        f: :handle_node_join,
        a: [hub, :_]
      }

      HookManager.register_handler(hub.storage.hook, Hook.pre_cluster_join(), handler)
    end
end
...
Let's also define the handler function!
defmodule CustomStrategy.Compare do
...
    def handle_node_join(hub, node) do
        IO.puts("Node #{node} is joining the hub #{hub.hub_id}")
    end
end
...
At this point we only have 2 more functions to implement. Let's continue by 
overriding the children_init/4 function which does nothing but return :ok.
If we return anything else other than :ok, the startup will fail for this
particular function.
...
@impl true
def children_init(_struct, _hub, _child_specs, _opts), do: :ok
...
We now have the last function to implement which is the belongs_to/4 function. This function will determine the node that the process should be distributed to. In our case, we will select a node based on its name. 
The very same function will also be used when processes are redistributed.
We also return a list of nodes although we only select one. The reason for this is that the function can return multiple nodes in case the process should be replicated.
For the sake of simplicity, we will ignore the replication factor in this example.
...
@impl true
def belongs_to(struct, %ProcessHub.Hub{} = hub, _child_id, _replication_factor) do
    hub_nodes = ProcessHub.Service.Cluster.nodes(hub.storage.misc, [:include_local])

    selected_node = case struct.direction do
        :asc -> Enum.sort(hub_nodes) |> Enum.at(0)
        :desc -> Enum.sort(hub_nodes) |> Enum.reverse() |> Enum.at(0)
    end

    [selected_node] # Ignore the replication factor and just return one node
end
...
This is it!

  
    
  
  Final implementation


Our final implementation of the CustomStrategy.Compare module should look like this:
defmodule CustomStrategy.Compare do
  alias ProcessHub.Strategy.Distribution.Base, as: DistributionStrategy

  # We can define custom options.
  defstruct [
    direction: :asc
  ]

  def handle_node_join(%ProcessHub.Hub{} = hub, node) do
      IO.puts("Node #{node} is joining the hub #{hub.hub_id}")
  end

  defimpl DistributionStrategy, for: CustomStrategy.Compare do
    alias ProcessHub.Service.HookManager
    alias ProcessHub.Constant.Hook

    @impl true
    def init(_struct, %ProcessHub.Hub{} = hub) do
      handler = %HookManager{
        id: :some_id_here,
        m: CustomStrategy.Compare,
        f: :handle_node_join,
        a: [hub, :_]
      }

      HookManager.register_handler(hub.storage.hook, Hook.pre_cluster_join(), handler)
    end

    @impl true
    def children_init(_struct, _hub, _child_specs, _opts), do: :ok

    @impl true
    def belongs_to(struct, %ProcessHub.Hub{} = hub, _child_ids, _replication_factor) do
        hub_nodes = ProcessHub.Service.Cluster.nodes(hub.storage.misc, [:include_local])

        Enum.map(child_ids, fn child_id ->
          selected_node = case struct.direction do
            :asc -> Enum.sort(hub_nodes) |> Enum.at(0)
            :desc -> Enum.sort(hub_nodes) |> Enum.reverse() |> Enum.at(0)
          end

          # Ignore the replication factor and just return one node
          {child_id, [selected_node]}
        end)
    end
  end
end

  
    
  
  Switching to the custom strategy


Replace the default distribution strategy with our custom strategy in the ProcessHub configuration.
defmodule MyApp.Application do
  use Application

  @impl true
  def start(_type, _args) do
    children = [process_hub()]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end

  defp process_hub() do
    {ProcessHub, %ProcessHub{
      hub_id: :my_hub,
      distribution_strategy: %CustomStrategy.Compare{
        direction: :desc
      }
    }}
  end
end


  
    
  
  Test drive


Let's start two nodes on separate terminals.
iex --name node1@127.0.0.1 --cookie mycookie -S mix

iex --name node2@127.0.0.1 --cookie mycookie -S mix

Connect the nodes by running the following command on the node1 terminal.
iex> Node.connect(:"node2@127.0.0.1")
true
Time to see the magic happen!
iex> ProcessHub.start_children(:my_hub, [
...>    %{id: "process1", start: {MyProcess, :start_link, [nil]}},
...>    %{id: "process2", start: {MyProcess, :start_link, [nil]}},
...>    %{id: "process3", start: {MyProcess, :start_link, [nil]}},
...>    %{id: "process4", start: {MyProcess, :start_link, [nil]}},
...>    %{id: "process5", start: {MyProcess, :start_link, [nil]}}
...> ])
{:ok, :start_initiated}
iex> ProcessHub.process_list(:my_hub, :global)
[
  {"process1", ["node2@127.0.0.1": #PID<23391.238.0>]},
  {"process2", ["node2@127.0.0.1": #PID<23391.239.0>]},
  {"process3", ["node2@127.0.0.1": #PID<23391.240.0>]},
  {"process4", ["node2@127.0.0.1": #PID<23391.241.0>]},
  {"process5", ["node2@127.0.0.1": #PID<23391.242.0>]}
]
We can confirm that all processes are started on the second node when we
pass the :desc option to the struct.

  Overview


The overall architecture of ProcessHub is designed to provide a reliable and resilient
distributed system that can handle network partitions, node failures, process migrations,
This is mostly achieved by asynchronous and non-blocking operations, and by using the
Supervisor behavior to monitor and restart processes when they die unexpectedly.
ProcessHub is eventually consistent meaning that it can guarantee that the state of the
system will eventually converge to a consistent state. This enables the system to be
scalable and highly available. 
ProcessHub internally uses a event-driven communication and ability to define listeners/hooks.
This enables some type of way to react to specific events such as process registration etc.
Most of the operations are carried out using special Task processes. These processes are
started on demand by the ProcessHub.Coordinator process and are supervised by the Task.Supervisor.
The coordinator process is responsible for coordinating the operations and making sure that
the operations are carried out in the correct order. This process is the heart of the system and
is responsible for the overall functionality of the system.

  
    
  
  Supervision tree


[image: supervision_tree]

  
    
  
  Processes


	coordinator - The coordinator process is responsible for coordinating the operations and making sure that the operations are carried out in the correct order. This process is the heart of the system and is responsible for the overall functionality of the system. 
All actions are dispatched to the coordinator process who then delegates the work to the correct handler process.

	distributed_supervisor - The distributed supervisor process is responsible for starting, stopping, and monitoring the processes in its local cluster. It uses the Supervisor behavior to monitor the processes and restart them when they die unexpectedly.

	event_queue_sup (external library) - The event queue supervisor starts and supervises the event queue  processes. The event queue is used to dispatch events within the Erlang distribution system to all nodes in the cluster. This provides a way to communicate between nodes and synchronize the operations in the system.
These processes are started by external library blockade.

	janitor - The janitor process is responsible for cleaning up the system and removing any stale data. It periodically checks the system for any stale data and removes it to keep the system clean and efficient.

	task_supervisor - The task supervisor process is responsible for supervising the task processes that are started on demand by the coordinator process. These processes are used to carry out the operations in the system and are supervised by the task supervisor process.

	worker_queue - The worker queue process is used to synchronize the operations that may introduce race conditions.



  
    
  
  Cluster Discovery and Formation


ProcessHub monitors connecting and disconnecting nodes and forms a cluster automatically
from the connected nodes that share the same hub_id. It's not required to start
the ProcessHub on all nodes in the cluster.

  
    
  
  Resilience and Reliability


ProcessHub uses the Supervisor behavior and leverages the features that come with it.
Each hub starts its own ProcessHub.DistributedSupervisor process, which is responsible for
starting, stopping, and monitoring the processes in its local cluster.
When a process dies unexpectedly, the ProcessHub.DistributedSupervisor will restart it
automatically.
ProcessHub also takes care of validating the child_spec before starting it and makes sure
it's started on the right node that the process belongs to.
If the process is being started on the wrong node, the initialization request will be forwarded
to the correct node.

  
    
  
  Locking Mechanism


ProcessHub utilizes the :blockade library to provide event-driven communication
and a locking mechanism.
It locks the local event queue by increasing its priority for some operations.
This allows the system to queue events and process them in order to preserve data integrity.
Other events can be processed once the priority level is set back to default.
To avoid deadlocks, the system places a timeout on the event queue priority and
restores it to its original value if the timeout is reached.


  

    
ProcessHub 
    



      
This is the main public API module for the ProcessHub library. It is recommended to use
only the functions defined in this module to interact with the ProcessHub library.
ProcessHub is a library that distributes processes within the BEAM cluster. It is designed to
be used as a building block for distributed applications that require process distribution
and synchronization across multiple nodes.

      


      
        Summary


  
    Types
  


    
      
        
  
    
    ProcessHub.Constant.Event - ProcessHub v0.4.0-beta
    
    

    


  
  

    
ProcessHub.Constant.Event 
    



      
Custom events defined as macros.

    ProcessHub.Constant.Hook - ProcessHub v0.4.0-beta
    
    

    


  
  

    
ProcessHub.Constant.Hook 
    



      
Defines the list of hooks that can be used to extend the functionality of ProcessHub.

ProcessHub.Constant.PriorityLevel 
    



      
Priority levels for the local event queue.

ProcessHub.Constant.StorageKey 

ProcessHub.Coordinator 
    



      
The ProcessHub coordinator module is responsible for coordinating most of the ProcessHub events and work.
In most cases, the coordinator module delegates the work to other service-based
modules or handler processes that are created on demand.
Each ProcessHub instance has its own coordinator process that handles the
coordination. These processes are supervised by the ProcessHub.Initializer
supervisor.
The coordinator stores state about the ProcessHub instance, such as the cluster nodes.
Additionally, the coordinator takes care of any periodic tasks required by the
ProcessHub instance, such as initial synchronization, propagation, etc.

    ProcessHub.DistributedSupervisor - ProcessHub v0.4.0-beta
    
    

    


  
  

    
ProcessHub.DistributedSupervisor behaviour
    



      
The ProcessHub distributed supervisor module is responsible for starting and stopping
the child processes distributed across the cluster.
Each ProcessHub instance has its own distributed supervisor that manages local
child processes.

ProcessHub.Future 
    



      
A struct representing an asynchronous operation that can be awaited for results.
The Future module provides functionality to handle asynchronous operations in ProcessHub,
allowing callers to wait for results from distributed process start/stop operations.
This is particularly useful when you need to verify the outcome of operations.

ProcessHub.Handler.ChildrenAdd.PostStartData 

ProcessHub.Handler.ChildrenAdd.StartHandle 
    



      
Handler for starting child processes.

ProcessHub.Handler.ChildrenAdd.SyncHandle 
    



      
Handler for synchronizing added child processes.

ProcessHub.Handler.ChildrenRem.StopHandle 
    



      
Handler for stopping child processes.

    ProcessHub.Handler.ChildrenRem.SyncHandle - ProcessHub v0.4.0-beta
    
    

    


  
  

    
ProcessHub.Handler.ChildrenRem.SyncHandle 
    



      
Handler for synchronizing stopped child processes.

ProcessHub.Handler.ClusterUpdate.NodeDown 
    



      
Handler for the node down event.

ProcessHub.Handler.ClusterUpdate.NodeUp 
    



      
Handler for the node up event.

ProcessHub.Handler.Synchronization.IntervalSyncHandle 
    



      
Handler for periodic synchronization.

ProcessHub.Handler.Synch